



International
Virtual
Observatory
Alliance

IVOA Astronomical Data Query Language

Version 1.5

IVOA Working Draft 2007 April 19

Deleted: *January*

Deleted: 26

This version:

1.5-20070419

Latest version:

<http://www.ivoa.net/Documents/latest/ADQL.html>

Previous version(s):

none

Editor(s):

Pedro Osuna and Inaki Ortiz

Author(s):

Inaki Ortiz, Jeff Lusted, Yuji Shirasaki, Maria A. Nieto-Santisteban, Masatoshi Ohishi, William O'Mullane, Alexander Szalay, Pedro Osuna, the VOQL-TEG and the VOQL Working Group.

Deleted: 1

Deleted: 26

Abstract

This document describes the Astronomical Data Query Language (ADQL). ADQL has been developed based on SQL92. This document describes the subset of the SQL grammar supported by ADQL. Special restrictions and extensions to

SQL92 have been defined in order to support generic and astronomy specific operations.

Status of This Document

This is a Working Draft. The first release of this document was 2006 January 22.

This is an IVOA Working Draft for review by IVOA members and other interested parties. It is a draft document and may be updated, replaced, or obsoleted by other documents at any time. It is inappropriate to use IVOA Working Drafts as reference materials or to cite them as other than “work in progress”.

A list of [current IVOA Recommendations and other technical documents](http://www.ivoa.net/Documents/) can be found at <http://www.ivoa.net/Documents/>.

Contents

1	Introduction	2
2	Astronomical Data Query Language (ADQL)	3
2.1	Characters, Keywords, Identifiers and Literals	3
2.1.1	Characters	3
2.1.2	Keywords and Identifiers	4
2.1.3	Literals	6
2.2	Query syntax	6
2.2.1	Table subqueries and Joins	7
2.2.2	Search condition	7
2.3	Functions	7
2.3.1	Mathematical Functions	8
2.3.2	Region	8
2.3.3	User Defined Functions	9
	Appendix A: BNF Grammar	10
	Appendix B: BNF Diagrams	19
	References	23

1 Introduction

The Astronomical Data Query Language (ADQL) is the language used by the International Virtual Observatory Alliance (IVOA) to represent astronomy queries posted to VO services. The IVOA has developed several standardized protocols to access astronomical data, e.g., SIAP and SSAP for image and spectral data respectively. These protocols might be satisfied using a single table query. However, different VO services have different needs in terms of query complexity and ADQL arises in this context.

Deleted: 1	Introduction	2
2	Astronomical Data Query Language (ADQL)	3
2.1	Characters, Keywords, Identifiers and Literals	3
2.1.1	Characters	3
2.1.2	Keywords and Identifiers	4
2.1.3	Literals	5
2.2	Query syntax	6
2.2.1	Subqueries and joins	6
2.2.2	Search condition	7
2.3	Functions	7
2.3.1	Mathematical Functions	7
2.3.2	Region	8
2.3.3	User Defined Functions	8
	Appendix A: BNF Grammar	9
	References	18

The ADQL specification pretends to avoid any distinction between core and advanced or extended functionalities. Hence ADQL has been built according to a single language definition (BNF based [1]). Any service making use of ADQL would then define the level of compliancy to the language. This would allow the notion of core and extension to be service-driven and it would decouple the language from the service specifications.

ADQL is based on the Structured Query Language (SQL), especially on SQL 92. The VO has a number of tabular data sets and many of them are stored in relational databases, making SQL a convenient access means. A subset of the SQL grammar has been extended to support queries which are specific to astronomy.

Deleted: specially

2 Astronomical Data Query Language (ADQL)

This section describes the ADQL language specification. We will define in subsequent sections the syntax for the special characters, reserved and non-reserved words, identifiers and literals and then, finally, the syntax for the query expression.

The formal notation for syntax of computing languages is often expressed in the “Backus Naur Form” BNF. This syntax is used by popular tools for producing parsers. Appendix A to this document provides the full BNF grammar for ADQL.

The following conventions are used through this document:

- Optional items are enclosed in meta symbols [and]
- A group of items is enclosed in meta symbols { and }
- Repetitive item (zero or more times) are followed by [...]
- Terminal symbols are enclosed by < and >
- Terminals of meta-symbol characters (=, [,], (,), <, >, *) are surrounded by quotes (“”) to distinguish them from meta-symbols
- Case insensitiveness otherwise stated.

2.1 Characters, Keywords, Identifiers and Literals

2.1.1 Characters

The language allows simple Latin letters (lower and upper case, i.e. {aA-zZ}), digits ({0-9}) and the following special characters:

- space

- single quote (')
- double quote (")
- percent (%)
- left and right parenthesis
- asterisk (*)
- plus sign (+)
- minus sign (-)
- comma (,)
- period (.)
- solidus (/)
- colon (:)
- less than operator (<)
- equals operator (=)
- greater than operator (>)
- underscore (_)

Deleted: .

2.1.2 Keywords and Identifiers

Besides the character set, the language provides a list of reserved keywords plus the syntax description for regular identifiers.

Deleted: and non-reserved

A reserved keyword has a special meaning in ADQL and cannot be used as an identifier. These keywords must be enforced and should be extensive as an escaping mechanism is already in place. We can extend the list of SQL92 reserved keywords to accommodate those useful for astronomical purposes and/or present in a subset of vendor specific languages only (e.g. TOP). This leads to the following list:

- SQL reserved keywords:

ABSOLUTE, ACTION, ADD, ALL, ALLOCATE, ALTER, AND, ANY, ARE, AS, ASC, ASSERTION, AT, AUTHORIZATION, AVG, BEGIN, BETWEEN, BIT, BIT LENGTH, BOTH, BY, CASCADE, CASCADED, CASE, CAST, CATALOG, CHAR, CHARACTER, CHARACTER LENGTH, CHAR LENGTH, CHECK, CLOSE, COALESCE, COLLATE, COLLATION, COLUMN, COMMIT, CONNECT, CONNECTION, CONSTRAINT, CONSTRAINTS, CONTINUE, CONVERT, CORRESPONDING, COUNT, CREATE, CROSS, CURRENT, CURRENT DATE, CURRENT TIME, CURRENT TIMESTAMP, CURRENT USER, CURSOR, DATE, DAY, DEALLOCATE, DEC, DECIMAL, DECLARE, DEFAULT, DEFERRABLE, DEFERRED, DELETE, DESC, DESCRIBE, DESCRIPTOR, DIAGNOSTICS, DISCONNECT, DISTINCT, DOMAIN, DOUBLE, DROP, ELSE, END, END-EXEC, ESCAPE, EXCEPT, EXCEPTION, EXEC, EXECUTE, EXISTS, EXTERNAL, EXTRACT, FALSE, FETCH, FIRST, FLOAT, FOR, FOREIGN, FOUND, FROM, FULL, GET, GLOBAL, GO, GOTO, GRANT, GROUP, HAVING, HOUR,

Comment [io1]: The list could be extended. For the time being I have just put the reserved keywords which are SQL92 compliant (plus some extensions)

UPLOAD is not present as we still have to discuss whether it is part of the language or the service

Formatted: Bullets and Numbering

Formatted: Indent: Left: 0.29"

IDENTITY, IMMEDIATE, IN, INDICATOR, INITIALLY, INNER, INPUT, INSENSITIVE, INSERT, INT, INTEGER, INTERSECT, INTERVAL, INTO, IS, ISOLATION, JOIN, KEY, LANGUAGE, LAST, LEADING, LEFT, LEVEL, LIKE, LOCAL, LOWER, MATCH, MAX, MIN, MINUTE, MODULE, MONTH, NAMES, NATIONAL, NATURAL, NCHAR, NEXT, NO, NOT, NULL, NULLIF, NUMERIC, OCTET LENGTH, OF, ON, ONLY, OPEN, OPTION, OR, ORDER, OUTER, OUTPUT, OVERLAPS, PAD, PARTIAL, POSITION, PRECISION, PREPARE, PRESERVE, PRIMARY, PRIOR, PRIVILEGES, PROCEDURE, PUBLIC, READ, REAL, REFERENCES, RELATIVE, RESTRICT, REVOKE, RIGHT, ROLLBACK, ROWS, SCHEMA, SCROLL, SECOND, SECTION, SELECT, SESSION, SESSION USER, SET, SIZE, SMALLINT, SOME, SPACE, SQL, SQLCODE, SQLERROR, SQLSTATE, SUBSTRING, SUM, SYSTEM USER, TABLE, TEMPORARY, THEN, TIME, TIMESTAMP, TIMEZONE HOUR, TIMEZONE MINUTE, TO, TRAILING, TRANSACTION, TRANSLATE, TRANSLATION, TRIM, TRUE, UNION, UNIQUE, UNKNOWN, UPDATE, UPPER, USAGE, USER, USING, VALUE, VALUES, VARCHAR, VARYING, VIEW, WHEN, WHENEVER, WHERE, WITH, WORK, WRITE, YEAR, ZONE

• ADQL reserved keywords:

ABS, ACOS, ASIN, ATAN, ATAN2, CEILING, COS, DEGREES, EXP, FLOOR, LOG, LOG10, MODE, PI, POWER, RADIANS, REGION, RAND, ROUND, SIN, SQRT, UDF, TAN, TOP, TRUNCATE,

The identifiers are used to express, for example, a table or a column reference name.

Both the identifiers and the keywords are case insensitive. They SHALL begin with a letter {aA-zZ}. Subsequent characters shall be letters, underscores or digits {0-9} as follows:

<Latin letter> [{ <underscore> | {<Latin letter> | <digit>} }

For practical purposes the language specification should be able to address reserved keyword and special character conflicts. To do so the language provides a way to escape a non-compliant identifier by using the double quote character as a delimiter.

ADQL allows making use of the same quoting mechanism to handle the case sensitiveness if needed.

Comment [io2]: These are mostly mathematical functions.

Formatted: Bullets and Numbering

Formatted: Indent: Left: 0.29"

Comment [io3]: Its syntax and usage will be defined in later sections

Jeff Lusted, through KA, proposed a region predicate construct in BNF

Comment [io4]: Placeholder for User Defined Functions or SQL macros. The usage of this keyword will be defined in the relevant section (2.3 Functions)

Still to discuss whether specific constructs like CROSS APPLY fit into this schema (AS)

Comment [io5]: Not SQL92 but useful nonetheless to be able to return a subset of records.

The is not a common naming convention across vendors: LIMIT and OFFSET are specific to MySQL and PostgreSQL, while ROWNUM is for Oracle, ROWCOUNT for Sybase and TOP is SQL Server specific.

I would propose to use TOP for historical reasons.

Deleted: A non-reserved keyword has a special meaning in specific contexts and can be used as an identifier in some other contexts.

Deleted: .

Deleted: ¶

Comment [io6]: These are mostly mathematical functions.

Comment [io7]: Its syntax and usage will be defined in later sections. Jeff Lusted, through KA, proposed a region predicate construct in BNF

Deleted: Reserved keywords are:¶ ALL, AND, AS, ASC, AVG, BETWEEN, BY, COUNT, DESC, DISTINCT, EXISTS, FALSE, FROM, FULL, GROUP, HAVING, IN, INNER, INTO, IS, JOIN, LEFT, LIKE, MAX, MIN, NATURAL, NOT, NULL, ON, OR, ORDER, OUTER, RIGHT, SELECT, SUM, TOP, TRUE, USING, WHERE. ¶

¶ Non-Reserved keywords are:¶ ABS, ACOS, ASIN, ATAN, ATAN2, CEILING, COS, DEGREES, EX ... [1]

Comment [io8]: PostgreSQL, Oracle and Sybase make use of a double quote delimiter though it is not standardized and some vendor ... [2]

2.1.3 Literals

Finally we define the syntax rules for the different data types: string, number and boolean.

Comment [io9]: Time interval taken out for the time being to make it simpler

Deleted: ,

Deleted: , date and time.

A string literal is a character expression delimited by single quotes.

Literal numbers are expressed in BNF as follows:

```
<unsigned numeric literal> ::=
    <exact numeric literal> | <approximate numeric literal>
<exact numeric literal> ::=
    <unsigned integer> [<period> [<unsigned integer>]]
    | <period><unsigned integer>
<unsigned integer> ::= <digit>
<approximate numeric literal> ::= <mantissa> E <exponent>
<mantissa> ::= <exact numeric literal>
<exponent> ::= <signed integer>
<signed integer> ::= [<sign>] <unsigned integer>
<sign> ::= <plus sign> | <minus sign>
```

Deleted: or

Comment [io10]: Open to discussion. The evolution of time-like literals should be agreed in a more general forum. Arnold rots did already give his inputs and we should agree on them first.

Deleted: The BNF for date and time literals constructs are:

```
¶
<date string> ::= <quote> <date value> <quote>¶
<date value> ::= <years value>
<minus sign> <months value> ¶
<minus sign> <days value>¶
<years value> ::= <datetime value>¶
<months value> ::= <date time value>¶
<days value> ::= <date time value>¶
<datetime value> ::= <unsigned integer>¶
¶
<time string> ::= <quote> <time value> [<time zone interval>] <quote>¶
<time value> ::= <hours value>
<colon> <minutes value> <colon>
<seconds value>¶
<hours value> ::= <datetime value>¶
<minutes value> ::= <datetime value>¶
<seconds value> ::= <seconds integer value> [ <period> [<seconds fraction>]]¶
<seconds integer value> ::=
<unsigned integer>¶
<seconds fraction> ::= <unsigned integer>¶
<time zone interval> ::= <sign>
<hours value> <colon> <minutes value>¶
¶
<timestamp string> ::= <quote> <date value> <space> <time value> [<time zone interval>] <quote>¶
```

A boolean literal expression is either TRUE, FALSE or UNKNOWN.

2.2 Query syntax

The compact syntax for the SELECT statement shows the main constructs for the query specification:

```
SELECT [ ALL | DISTINCT ]
    [ TOP <unsigned integer> ]
    { * | COUNT(*) | { <expression> [ AS <name> ] } [,...] }
    [ INTO <target specification> ]
FROM <table reference> [ <alias> ] [,...]
[ { [NATURAL] { LEFT | RIGHT } [OUTER] } | INNER | FULL
    [OUTER] } JOIN <alias> { ON <search condition> | USING(col1,
    col2,...) } ]
[ WHERE <search condition> ]
[ GROUP BY <column> [,...] ]
[ HAVING <search condition> ]
[ ORDER BY { <column> | <unsigned integer> } [ ASC | DESC ]
    [,...]]
```

Comment [io11]: Not present in the BNF. A point of future debate. We could even delete it temporarily.

Deleted:

Comment [io12]: Requires further description. Section 2.2.3

The SELECT statement defines a query to some derived table(s) specified in the FROM clause. As a result of this query, a subset of the table(s) is returned. The order of the rows MAY be arbitrary unless ORDER BY clause is specified. The

order of the columns to return SHOULD be the same as the order specified in the selection list, or the order defined in the original table if asterisk is specified.

Deleted: as

Deleted: d

TOP n construct is used to return the first n-rows.

Comment [io13]: What about OFFSET

The selection list MAY include any numeric, string value expression.

Comment [io14]: Date/Time usage should be discussed (see comment end of section on Literals.

In the following sections some constructs requiring further description are presented.

Deleted: and datetime

2.2.1 Table subqueries and Joins

Deleted: S

Deleted: ies

Deleted: j

Table subqueries are present and can be used by some existing predicates within the search condition (IN and BETWEEN most likely) or as an artifact of building derived tables.

Deleted: A

Deleted: y

Comment [io15]: This is useful if we finally go for a Region construct as a form of derived table

Among the different types of joins ADQL supports qualified ones only. These are INNER and OUTER ones (LEFT, RIGHT and FULL). All of these can be NATURAL or not. The join condition does not support embedded sub joins.

Deleted: can contain a simple query description or a join statement

Comment [io16]: What about CROSS JOIN?

2.2.2 Search condition

The search condition can be part of several other clauses: JOIN, HAVING and, obviously, WHERE. Standard logical operators are present in its description (AND, OR and NOT). Six different types of predicates are present in which different types of reserved keywords or characters are used:

- Standard comparison operators: =, !=, <>, <, >, <=, >=
- BETWEEN
- LIKE
- NULL
- EXISTS

[The full syntax for the SELECT statement is present in Appendix A.]

2.3 Functions

Comment [io17]: To define explicitly where to use these functions.

ADQL declares a list of non reserved keywords (section 2.1.2) which defines a set of special functions to enhance the astronomical usage of the language.

These can be split into three different types:

We agreed that REGION should be part of the WHERE clause only (VOQL-TEG Meeting #2)

2.3.1 Mathematical Functions

The next table shows the description of the mathematical functions.

Name	Return type	Comment
acos(x)	double	Basic function. Inverse cosine.
asin(x)	double	Basic function. Inverse sine.
atan(x)	double	Basic function. Inverse tangent.
atan2(x,y)	double	Basic function. Inverse tangent of x/y.
cos(x)	double	Basic function. Cosine.
sin(x)	double	Basic function. Sine.
tan(x)	double	Basic function. Tangent.
abs(x)	double	Basic function. Absolute value.
ceiling(x)	double	Basic function. Smallest integer not less than argument.
degrees(x)	double	Basic function. Radians to degrees.
exp(x)	double	Basic function. Exponential.
floor(x)	double	Basic function. Larger integer not greater than argument.
log(x)	double	Basic function. Natural logarithm.
log10(x)	double	Basic function. Base 10 logarithm.
mod(x, y)	double	Basic function. Remainder of y/x.
pi()	double	Basic function. Pi constant.
power(x, y)	double	Basic function. X raised to the power of Y.
radians(x)	double	Basic function. Degree to radians.
sqrt(x)	double	Basic function. Square root.
rand(x)	double	Basic function. Random value between 0.0 and 1.0. <u>It can take a seed value.</u>
round(x, n)	double	Basic function. Round to nearest integer.
truncate(x, n)	double	Basic function. Truncate to n decimal places.

2.3.2 Region

To be discussed.

Comment [jo18]: We just agreed that the REGION construct was supposed to be part of the WHERE clause.

KA sent some comments from Jeff Lusted about the possible BNF syntax for the region predicate.

Full support of the STC is an issue

2.3.3 User Defined Functions

To be discussed.

Comment [io19]: Or SQL macros... which name should we choose?

We still have to see whether for example the CROSS APPLY could fit in the UDF container.

Comment [io20]: Missing stuff:
SELECT INTO detailed syntax
Usage of Comments –
ADQL Version
UPLOAD to be discussed if it's part of the language
Qualified table names (through service identifiers)
XPath?

Appendix A: BNF Grammar

Comment [io21]: This is a preliminary and incomplete version of the grammar. It should only be used as a starting point.

SELECT INTO construct missing

SQL macros, User Define Functions and Region constructs are not reflected in this preliminary attempt either.

<ADQL language character> ::=

<simple Latin letter>
| <digit>
| <SQL special character>

<ADQL reserved word> ::=

ABS
| ACOS
| ASIN
| ATAN
| ATAN2
| CEILING
| COS
| DEGREES
| EXP
| FLOOR
| LOG
| LOG10
| MODE
| PI
| POWER
| RADIANS
| REGION
| RAND
| ROUND
| SIN
| SQRT
| TOP
| UDF
| TAN
| TRUNCATE

<SQL reserved word> ::=

ABSOLUTE | ACTION | ADD | ALL | ALLOCATE | ALTER | AND | ANY | ARE
| AS | ASC | ASSERTION | AT | AUTHORIZATION | AVG
| BEGIN | BETWEEN | BIT | BIT_LENGTH | BOTH | BY
| CASCADE | CASCADED | CASE | CAST | CATALOG | CHAR | CHARACTER
| CHARACTER_LENGTH
| CHAR_LENGTH | CHECK | CLOSE | COALESCE | COLLATE | COLLATION
| COLUMN | COMMIT
| CONNECT | CONNECTION | CONSTRAINT | CONSTRAINTS | CONTINUE
| CONVERT | CORRESPONDING | COUNT
| CREATE | CROSS | CURRENT | CURRENT_DATE | CURRENT_TIME
| CURRENT_TIMESTAMP
| CURRENT_USER | CURSOR
| DATE | DAY | DEALLOCATE | DEC | DECIMAL | DECLARE | DEFAULT
| DEFERRABLE | DEFERRED | DELETE | DESC | DESCRIBE | DESCRIPTOR
| DIAGNOSTICS
| DISCONNECT | DISTINCT | DOMAIN | DOUBLE | DROP
| ELSE | END | END-EXEC | ESCAPE | EXCEPT | EXCEPTION | EXEC

Deleted: -- ¶

<ADQL language character> ::= ¶
<simple Latin letter> ¶
| <digit> ¶
| <ADQL special character> ¶

<simple Latin letter> ::= ¶
<simple Latin upper case letter> ¶
| <simple Latin lower case letter> ¶

<simple Latin upper case letter> ::= ¶
A | B | C | D | E | F | G | H | I | J | K |
L | M | N | O | P | Q | R | S | T | U | V |
W | X | Y | Z ¶

<simple Latin lower case letter> ::= ¶
a | b | c | d | e | f | g | h | i | j | k | l | m |
n | o | p | q | r | s | t | u | v | w | x | y |
z ¶

<digit> ::= ¶
0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9 ¶

<ADQL special character> ::= ¶

<space> ¶
| <double quote> ¶
| <percent> ¶
| <quote> ¶
| <left paren> ¶
| <right paren> ¶
| <asterisk> ¶
| <plus sign> ¶
| <comma> ¶
| <minus sign> ¶
| <period> ¶
| <solidus> ¶
| <colon> ¶
| <less than operator> ¶
| <greater than operator> ¶
| <equals operator> ¶
| <underscore> ¶

<space> ::= !! space character in
character set in use ¶

<double quote> ::= " ¶

<percent> ::= % ¶

<quote> ::= ' ¶

<left paren> ::= (¶

<right paren> ::=) ¶

<asterisk> ::= * ¶

<plus sign> ::= + ¶

... [3]

Formatted: Font: (Default) Arial

| EXECUTE
| EXISTS | EXTERNAL | EXTRACT
| FALSE | FETCH | FIRST | FLOAT | FOR | FOREIGN | FOUND | FROM | FULL
| GET | GLOBAL | GO | GOTO | GRANT | GROUP
| HAVING | HOUR
| IDENTITY | IMMEDIATE | IN | INDICATOR | INITIALLY | INNER | INPUT
| INSENSITIVE
| INSERT | INT | INTEGER | INTERSECT | INTERVAL | INTO | IS
| ISOLATION
| JOIN
| KEY
| LANGUAGE | LAST | LEADING | LEFT | LEVEL | LIKE | LOCAL | LOWER
| MATCH | MAX | MIN | MINUTE | MODULE | MONTH
| NAMES | NATIONAL | NATURAL | NCHAR | NEXT | NO | NOT | NULL
| NULLIF | NUMERIC
| OCTET_LENGTH | OF | ON | ONLY | OPEN | OPTION | OR | ORDER | OUTER
| OUTPUT | OVERLAPS
| PAD | PARTIAL | POSITION | PRECISION | PREPARE | PRESERVE | PRIMARY
| PRIOR | PRIVILEGES | PROCEDURE | PUBLIC
| READ | REAL | REFERENCES | RELATIVE | RESTRICT | REVOKE | RIGHT
| ROLLBACK | ROWS
| SCHEMA | SCROLL | SECOND | SECTION | SELECT | SESSION
| SESSION_USER | SET
| SIZE | SMALLINT | SOME | SPACE | SQL | SQLCODE | SQLERROR
| SQLSTATE | SUBSTRING | SUM | SYSTEM_USER
| TABLE | TEMPORARY | THEN | TIME | TIMESTAMP | TIMEZONE_HOUR
| TIMEZONE_MINUTE
| TO | TRAILING | TRANSACTION | TRANSLATE | TRANSLATION | TRIM | TRUE
| UNION | UNIQUE | UNKNOWN | UPDATE | UPPER | USAGE | USER | USING
| VALUE | VALUES | VARCHAR | VARYING | VIEW
| WHEN | WHENEVER | WHERE | WITH | WORK | WRITE
| YEAR
| ZONE

<SQL special character> ::=

<space>
| <double quote>
| <percent>
| <quote>
| <left paren>
| <right paren>
| <asterisk>
| <plus sign>
| <comma>
| <minus sign>
| <period>
| <solidus>
| <colon>
| <less than operator>
| <greater than operator>
| <equals operator>
| <underscore>

<approximate numeric literal> ::= <mantissa> E <exponent>

<as clause> ::= [AS] <column_name>

<asterisk> ::= *

<between_predicate> ::=
<value_expression> [NOT] BETWEEN <value_expression> AND <value_expression>

<boolean_factor> ::= [NOT] <boolean_test>

<boolean_literal> ::= TRUE | FALSE

<boolean_primary> ::=
<predicate> | <left_paren> <search_condition> <right_paren>

<boolean_term> ::= <boolean_factor> | <boolean_term> AND <boolean_factor>

<boolean_test> ::= <boolean_primary> [IS [NOT] <truth_value>]

<character_factor> ::= <character_primary>

<character_primary> ::= <value_expression_primary>

<character_representation> ::= <nonquote_character> | <quote_symbol>

<character_string_literal> ::=
<quote> [<character_representation>...] <quote>
[{ <separator>... <quote> [<character_representation>...] <quote> }...]

Formatted: Font: (Default) Arial

<character_value_expression> ::= <concatenation> | <character_factor>

<colon> ::= :

<column_name> ::= <identifier>

<column_name_list> ::= <column_name> [{ <comma> <column_name> }...]

<column_reference> ::= [<qualifier> <period>] <column_name>

<comma> ::= ,

<comment> ::= <comment_introducer> [<comment_character>...] <newline>

<comment_character> ::= <nonquote_character> | <quote>

<comment_introducer> ::= <minus_sign><minus_sign> [<minus_sign>...]

<comparison_operator> ::=
<equals_operator>
| <not_equals_operator>
| <less_than_operator>
| <greater_than_operator>
| <less_than_or_equals_operator>
| <greater_than_or_equals_operator>

<comparison_predicate> ::= <value_expression> <comparison_operator> <value_expression>

<concatenation> ::=

<character_value_expression> <concatenation_operator> <character_factor>

<concatenation_operator> ::= ||

<correlation_name> ::= <identifier>

<correlation_specification> ::= [AS] <correlation_name>

<delimited_identifier> ::=
<double_quote> <regular_identifier> <double_quote>

<delimiter_token> ::=
<character_string_literal>
| <boolean_literal>
| <SQL_special_character>
| <not_equals_operator>
| <greater_than_or_equals_operator>
| <less_than_or_equals_operator>
| <concatenation_operator>
| <double_period>
| <left_bracket>
| <right_bracket>

<derived_column> ::= <value_expression> [<as_clause>]

<digit> ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

<double_period> ::= ..

<double_quote> ::= "

<equals_operator> ::= =

<exact_numeric_literal> ::=
<unsigned_integer> [<period> [<unsigned_integer>]]
| <period> <unsigned_integer>

<exists_predicate> ::= EXISTS <table_subquery>

<exponent> ::= <signed_integer>

<factor> ::= [<sign>] <numeric_primary>

<from_clause> ::= FROM <table_reference> [{ <comma> <table_reference> }...]

<general_set_function> ::=
<set_function_type> <left_paren> [<set_quantifier>] <value_expression> <right_paren>

<greater_than_operator> ::= >

<greater_than_or_equals_operator> ::= >=

<group_by_clause> ::= GROUP BY <grouping_column_reference_list>

<grouping_column_reference> ::= <column_reference>

<grouping_column_reference_list> ::=
<grouping_column_reference> [{ <comma> <grouping_column_reference> } ...]

<having_clause> ::= HAVING <search_condition>

<identifier> ::= <regular_identifier> | <delimited_identifier>

<in_predicate> ::= <value_expression> [NOT] IN <in_predicate_value>

<in_predicate_value> ::=
<table_subquery> | <left_paren> <in_value_list> <right_paren>

<in_value_list> ::=
<value_expression> { <comma> <value_expression> } ...

<join_column_list> ::= <column_name_list>

<join_condition> ::= ON <search_condition>

<join_specification> ::= <join_condition> | <named_columns_join>

<join_type> ::=
INNER
| <outer_join_type> [OUTER]

<joined_table> ::=
<qualified_join>
| <left_paren> <joined_table> <right_paren>

<keyword> ::= <SQL_reserved_word> | <ADQL_reserved_word>

<left_bracket> ::= [

<left_paren> ::= (

<less_than_operator> ::= <

<less_than_or_equals_operator> ::= <=

<like_predicate> ::=
<character_value_expression> [NOT] LIKE <character_value_expression>

<mantissa> ::= <exact_numeric_literal>

<math_function> ::=
ABS <left_paren> <numeric_value_expression> <right_paren>
| CEILING <left_paren> <numeric_value_expression> <right_paren>
| DEGREES <left_paren> <numeric_value_expression> <right_paren>
| EXP <left_paren> <numeric_value_expression> <right_paren>
| FLOOR <left_paren> <numeric_value_expression> <right_paren>
| LOG <left_paren> <numeric_value_expression> <right_paren>
| PI <left_paren> <right_paren>
| POWER <left_paren> <numeric_value_expression> <comma> <unsigned_integer>
<right_paren>
| RADIANS <left_paren> <numeric_value_expression> <right_paren>
| SQUARE <left_paren> <numeric_value_expression> <right_paren>

| SQRT <left_paren> <numeric_value_expression> <right_paren>
| LOG10 <left_paren> <numeric_value_expression> <right_paren>
| RAND <left_paren> [<numeric_value_expression>] <right_paren>
| ROUND <left_paren> <numeric_value_expression> <right_paren>
| TRUNCATE <left_paren> <numeric_value_expression> <right_paren>

<minus_sign> ::= -

<named_columns_join> ::= USING <left_paren> <join_column_list> <right_paren>

<newline> ::= !! implementation defined end of line indicator

<nondelimiter_token> ::=
 <regular_identifier>
 | <keyword>
 | <unsigned_numeric_literal>

<nonquote_character> ::= !! One ASCII character

<not_equals_operator> ::= <> | !=

<null_predicate> ::= IS [NOT] NULL

<numeric_primary> ::= <value_expression_primary> | <numeric_value_function>

<numeric_value_expression> ::=
 <term>
 | <numeric_value_expression> <plus_sign> <term>
 | <numeric_value_expression> <minus_sign> <term>

<numeric_value_function> ::= <trigonometric_function> | <math_function>

<order_by_clause> ::= ORDER BY <sort_specification_list>

<ordering_specification> ::= ASC | DESC

<outer_join_type> ::= LEFT | RIGHT | FULL

<percent> ::= %

<period> ::= .

<plus_sign> ::= +

<predicate> ::=
 <comparison_predicate>
 | <between_predicate>
 | <in_predicate>
 | <like_predicate>
 | <null_predicate>
 | <exists_predicate>
 | <region_predicate>

<qualified_join> ::=
 <table_reference> [NATURAL] [<join_type>] JOIN <table_reference> [<join_specification>]

<qualifier> ::= <table_name> | <correlation_name>

<query_specification> ::=

SELECT [<set_quantifier>] [<set_limit>] <select_list> <table_expression>

<quote> ::= '

<quote symbol> ::= <quote> <quote>

<region_predicate> ::=

<table_reference> <period> REGION

<left_paren>

<region_specification>

[<comma> <column_reference> [{ <comma> <column_reference> } ...]

<right_paren>

<region_specification> ::=

!! Region definition from STC schema here

<regular_identifier> ::=

<simple_Latin_letter>... [{ <underscore> | <simple_Latin_letter> | <digit> } ...]

Formatted: Font: (Default) Arial

<right_bracket> ::=]

<right_paren> ::=)

<schema_name> ::= [<identifier> <period>] <identifier>

<search_condition> ::=

<boolean_term>

| <search_condition> OR <boolean_term>

<select_list> ::=

<asterisk>

| <select_sublist> [{ <comma> <select_sublist> } ...]

<select_sublist> ::= <derived_column> | <qualifier> <period> <asterisk>

<separator> ::= { <comment> | <space> | <newline> } ...

<set_function_specification> ::=

COUNT <left_paren> <asterisk> <right_paren>

| <general_set_function>

<set_function_type> ::= AVG | MAX | MIN | SUM | COUNT

<set_limit> ::= TOP <unsigned_integer>

<set_quantifier> ::= DISTINCT | ALL

<sign> ::= <plus_sign> | <minus_sign>

<signed_integer> ::= [<sign>] <unsigned_integer>

<simple_Latin_letter> ::=

<simple_Latin_upper_case_letter>

| <simple Latin lower case letter>

<simple Latin lower case letter> ::=
a | b | c | d | e | f | g | h | i | j | k | l | m | n
| o | p | q | r | s | t | u | v | w | x | y | z

<simple Latin upper case letter> ::=
A | B | C | D | E | F | G | H | I | J | K | L | M | N
| O | P | Q | R | S | T | U | V | W | X | Y | Z

<solidus> ::= /

<sort key> ::= <column_name> | <unsigned_integer>

<sort specification> ::= <sort key> [<ordering specification>]

<sort specification list> ::= <sort specification> [{ <comma> <sort specification> } ...]

<space> ::= !! space character here

<string value expression> ::= <character value expression>

<table expression> ::=
<from clause>
[<where clause>]
[<group by clause>]
[<having clause>]
[<order by clause>]

<table_name> ::= [<schema_name> <period>] <identifier>

<table reference> ::=
<table_name> [<correlation specification>]
| <table_subquery> <correlation specification>
| <joined table>

<table subquery> ::=
<left paren> <query specification> <right paren>

<term> ::=
<factor>
| <term> <asterisk> <factor>
| <term> <solidus> <factor>

<token> ::=
<nondelimiter token>
| <delimiter token>

<trigonometric function> ::=
ACOS <left paren> <numeric value expression> <right paren>
| ASIN <left paren> <numeric value expression> <right paren>
| ATAN <left paren> <numeric value expression> <right paren>
| ATAN2 <left paren> <numeric value expression> <comma> <numeric value expression>
<right paren>
| COS <left paren> <numeric value expression> <right paren>
| COT <left paren> <numeric value expression> <right paren>

| SIN <left paren> <numeric value expression> <right paren>

| TAN <left paren> <numeric value expression> <right paren>

<truth_value> ::= TRUE | FALSE | UNKNOWN

<underscore> ::= _

<unsigned integer> ::= <digit> ...

<unsigned_literal> ::= <unsigned_numeric_literal>

<unsigned_numeric_literal> ::=

<exact_numeric_literal>

| <approximate_numeric_literal>

<unsigned_value_specification> ::= <unsigned_literal>

<value_expression> ::=

<numeric_value_expression>

| <string_value_expression>

<value_expression_primary> ::=

<unsigned_value_specification>

| <column_reference>

| <set_function_specification>


| <left paren> <value_expression> <right paren>


<where_clause> ::= WHERE <search_condition>


Appendix B: BNF Diagrams


Formatted: Normal

Below there are several diagrams showing the dependencies among all non-terminal symbols of the grammar present in Appendix A.









References

[1] BNF Grammar for ISO/IEC 9075:1992 – Database Language SQL (SQL-92)
<http://savage.net.au/SQL/sql-92.bnf.html>

Deleted: [2] BNF for SQL92
<http://www.sqlzoo.net/sql92.html>

Reserved keywords are:

ALL, AND, AS, ASC, AVG, BETWEEN, BY, COUNT, DESC, DISTINCT, EXISTS, FALSE, FROM, FULL, GROUP, HAVING, IN, INNER, INTO, IS, JOIN, LEFT, LIKE, MAX, MIN, NATURAL, NOT, NULL, ON, OR, ORDER, OUTER, RIGHT, SELECT, SUM, TOP, TRUE, USING, WHERE.

Non-Reserved keywords are^[io1]:

ABS, ACOS, ASIN, ATAN, ATAN2, CEILING, COS, DEGREES, EXP, FLOOR, LOG, LOG10, MODE, PI, POWER, RADIANS, REGION^[io2], RAND, ROUND, SIN, SQRT, UDF, TAN, TRUNCATE.

PostgreSQL, Oracle and Sybase make use of a double quote delimiter though it is not standardized and some vendors do support single quote as well.

--

<ADQL language character> ::=

```

    <simple Latin letter>
  |
    <digit>
  |
    <ADQL special character>

```

<simple Latin letter> ::=

```

    <simple Latin upper case letter>
  |
    <simple Latin lower case letter>

```

<simple Latin upper case letter> ::=

A | B | C | D | E | F | G | H | I | J | K | L | M | N | O | P | Q | R | S | T | U | V | W | X | Y | Z

<simple Latin lower case letter> ::=

a | b | c | d | e | f | g | h | i | j | k | l | m | n | o | p | q | r | s | t | u | v | w | x | y | z

<digit> ::=

0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

<ADQL special character> ::=

```

    <space>
  |
    <double quote>
  |
    <percent>
  |
    <quote>
  |
    <left paren>
  |
    <right paren>
  |
    <asterisk>
  |
    <plus sign>
  |
    <comma>
  |
    <minus sign>
  |
    <period>
  |
    <solidus>
  |
    <colon>
  |
    <less than operator>
  |
    <greater than operator>
  |
    <equals operator>
  |
    <underscore>

```

<space> ::= !! space character in character set in use

<double quote> ::= "

<percent> ::= %

<quote> ::= '

<left paren> ::= (

<right paren> ::=)

<asterisk> ::= *

<plus sign> ::= +

<comma> ::= ,

<minus sign> ::= -

<period> ::= .

<solidus> ::= /

<colon> ::= :

<less than operator> ::= <

<equals operator> ::= =

<greater than operator> ::= >

<underscore> ::= _

--

<token> ::=

| <nondelimiter token>
| <delimiter token>

<nondelimiter token> ::=

| <regular identifier>
| <keyword>
| <unsigned numeric literal>
| <bit string literal>
| <hex string literal>

<regular identifier> ::= <identifier body>

<identifier body> ::= <identifier start> [{ <underscore> | <identifier part> } ...]

<identifier start> ::= <simple Latin letter>

<identifier part> ::= <identifier start> | <digit>

<keyword> ::= <reserved word> | <non-reserved word>

<reserved word> ::=

ALL | AND | AS | ASC | AVG | BETWEEN | BY | COUNT | DESC | DISTINCT | EXISTS
| FALSE | FROM | FULL | GROUP | HAVING | IN | INNER | INTO | IS | JOIN | LEFT |
LIKE | MAX | MIN | NATURAL | NOT | NULL | ON | OR | ORDER | OUTER | RIGHT |
SELECT | SUM | TOP | TRUE | USING | WHERE

<non-reserved word> ::=

ABS | ACOS | ASIN | ATAN | ATAN2 | CEILING | COS | DEGREES | EXP | FLOOR |
LOG | LOG10 | MODE | PI | POWER | RADIANS | REGION | RAND | ROUND | SIN |
SQRT | UDF | TAN | TRUNCATE

<unsigned numeric literal> ::=

<exact numeric literal>
| <approximate numeric literal>

<exact numeric literal> ::=

<unsigned integer> [<period> [<unsigned integer>]]
| <period> <unsigned integer>

<unsigned integer> ::= <digit> ...

<approximate numeric literal> ::=

<mantissa> E <exponent>

<mantissa> ::= <exact numeric literal>

<exponent> ::= <signed integer>

<signed integer> ::= [<sign>] <unsigned integer>

<sign> ::= <plus sign> | <minus sign>

<character representation> ::= <nonquote character> | <quote symbol>

<nonquote character> ::= !! See the Syntax rules

<quote symbol> ::= <quote> <quote>

<separator> ::= { <comment> | <space> | <newline> }...

<comment> ::= <comment introducer> [<comment character>...] <newline>

<comment introducer> ::= <minus sign><minus sign> [<minus sign>...]

<comment character> ::= <nonquote character> | <quote>

<newline> ::= !! implementation defined end of line indicator

<bit string literal> ::=

B <quote> [<bit> ...] <quote> [{ <separator>... <quote> [<bit>...] <quote> }...]

<bit> ::= 0 | 1

<hex string literal> ::=

X <quote> [<hexit> ...] <quote> [{ <separator>... <quote> [<hexit>...] <quote>
}...]

<hexit> ::= <digit> | A | B | C | D | E | F | a | b | c | d | e | f

--

<delimiter token> ::=

| <character string literal>
| <boolean literal>
| <date string>
| <time string>
| <timestamp string>
| <ADQL special character>
| <not equals operator>
| <greater than or equals operator>
| <less than or equals operator>
| <concatenation operator>
| <double period>
| <left bracket>
| <right bracket>

<character string literal> ::= <quote> [<character representation>...] <quote> [{ <separator>... <quote> [<character representation>...] <quote> }...]

<boolean literal> ::= TRUE | FALSE

<date string> ::= <quote> <date value> <quote>

<date value> ::= <years value> <minus sign> <months value> <minus sign> <days value>

<years value> ::= <datetime value>

<datetime value> ::= <unsigned integer>

<months value> ::= <datetime value>

<days value> ::= <datetime value>

<time string> ::= <quote> <time value> [<time zone interval>] <quote>

<time value> ::= <hours value> <colon> <minutes value> <colon> <seconds value>

<hours value> ::= <datetime value>

<minutes value> ::= <datetime value>

<seconds value> ::= <seconds integer value> [<period> [<seconds fraction>]]

<seconds integer value> ::= <unsigned integer>

<seconds fraction> ::= <unsigned integer>

<time zone interval> ::= <sign> <hours value> <colon> <minutes value>

<timestamp string> ::= <quote> <date value> <space> <time value> [<time zone interval>] <quote>

<not equals operator> ::= <>

<greater than or equals operator> ::= >=

<less than or equals operator> ::= <=

<concatenation operator> ::= ||

<double period> ::= ..

<left bracket> ::= [

<right bracket> ::=]

--

<search condition> ::=
 <boolean term> | <search condition> OR <boolean term>

<boolean term> ::=
 <boolean factor> | <boolean term> AND <boolean factor>

<boolean factor> ::= [NOT] <boolean test>

<boolean test> ::= <boolean primary> [IS [NOT] <truth value>]

<boolean primary> ::= <predicate> | <left paren> <search condition> <right paren>

<predicate> ::=
 <comparison predicate>
 | <between predicate>
 | <in predicate>
 | <like predicate>
 | <null predicate>
 | <exists predicate>

<comparison predicate> ::= <row value constructor> <comp op> <row value constructor>

<row value constructor> ::=
 <row value constructor element>
 | <left paren> <row value constructor list> <right paren>
 | <row subquery>

<row value constructor element> ::=
 <value expression>
 | <null specification>

<value expression> ::=
 <numeric value expression>
 | <string value expression>
 | <datetime value expression>
 | <interval value expression>

<numeric value expression> ::=
 <term>
 | <numeric value expression> <plus sign> <term>
 | <numeric value expression> <minus sign> <term>

<term> ::=
 <factor>

```

    | <term> <asterisk> <factor>
    | <term> <solidus> <factor>

<factor> ::= [ <sign> ] <numeric primary>

<numeric primary> ::= <value expression primary> | <numeric value function>

<value expression primary> ::=
    <unsigned value specification>
    | <column reference>
    | <set function specification>
    | <scalar subquery>
    | <left paren> <value expression> <right paren>

<unsigned value specification> ::= <unsigned literal>
<unsigned literal> ::= <unsigned numeric literal>

<column reference> ::= [ <qualifier> <period> ] <column name>

<qualifier> ::= <table name> | <correlation name>

<correlation name> ::= <identifier>

<set function specification> ::=
    COUNT <left paren> <asterisk> <right paren>
    | <general set function>

<general set function> ::=
    <set function type> <left paren> [ <set quantifier> ] <value expression> <right paren>

<set function type> ::= AVG | MAX | MIN | SUM | COUNT

<set quantifier> ::= DISTINCT | ALL

--
<scalar subquery> ::= <subquery>

<subquery> ::= <left paren> <query expression> <right paren>

<query expression> ::= <non-join query expression> | <joined table>

<non-join query expression> ::=
    <non-join query term>

<non-join query term> ::=
    <non-join query primary>

<non-join query primary> ::= <simple table> | <left paren> <non-join query expression> <right
paren>

<simple table> ::=
    <query specification>

<query specification> ::=
    SELECT [ <set quantifier> ] <select list> <table expression>

```

```

<select list> ::=
    <asterisk>
    | <select sublist> [ { <comma> <select sublist> }... ]

<select sublist> ::= <derived column> | <qualifier> <period> <asterisk>

<derived column> ::= <value expression> [ <as clause> ]

<as clause> ::= [ AS ] <column name>

<table expression> ::=
    <from clause>
    [ <where clause> ]
    [ <group by clause> ]
    [ <having clause> ]
    [ <order by clause> ]

<from clause> ::= FROM <table reference> [ { <comma> <table reference> }... ]

<table reference> ::=
    <table name> [ <correlation specification> ]
    | <derived table> <correlation specification>
    | <joined table>

<correlation specification> ::=
    [ AS ] <correlation name> [ <left paren> <derived column list> <right paren> ]

<derived column list> ::= <column name list>

<derived table> ::= <table subquery>

<table subquery> ::= <subquery>

<joined table> ::=
    <cross join>
    | <qualified join>
    | <left paren> <joined table> <right paren>

<cross join> ::=
    <table reference> CROSS JOIN <table reference>

<qualified join> ::=
    <table reference> [ NATURAL ] [ <join type> ] JOIN <table reference> [ <join
specification> ]

<join type> ::=
    INNER
    | <outer join type> [ OUTER ]

<outer join type> ::= LEFT | RIGHT | FULL

<join specification> ::= <join condition> | <named columns join>

<join condition> ::= ON <search condition>

```

<named columns join> ::= USING <left paren> <join column list> <right paren>
 <join column list> ::= <column name list>
 <where clause> ::= WHERE <search condition>
 <group by clause> ::= GROUP BY <grouping column reference list>
 <grouping column reference list> ::=
 <grouping column reference> [{ <comma> <grouping column reference> }...]
 <grouping column reference> ::= <column reference>
 <having clause> ::= HAVING <search condition>
 <query term> ::= <non-join query term> | <joined table>
 <query primary> ::= <non-join query primary> | <joined table>
 --
 <character value expression> ::= <concatenation> | <character factor>
 <concatenation> ::= <character value expression> <concatenation operator> <character factor>
 <character factor> ::= <character primary>
 <character primary> ::= <value expression primary> |
 <datetime value expression> ::=
 <datetime term>
 | <interval value expression> <plus sign> <datetime term>
 | <datetime value expression> <plus sign> <interval term>
 | <datetime value expression> <minus sign> <interval term>
 <interval term> ::=
 <interval factor>
 | <interval term 2> <asterisk> <factor>
 | <interval term 2> <solidus> <factor>
 | <term> <asterisk> <interval factor>
 <interval factor> ::= [<sign>] <interval primary>
 <interval primary> ::= <value expression primary> [<interval qualifier>]
 <interval term 2> ::= <interval term>
 <interval value expression> ::=
 <interval term>
 | <interval value expression 1> <plus sign> <interval term 1>
 | <interval value expression 1> <minus sign> <interval term 1>
 | <left paren> <datetime value expression> <minus sign> <datetime term> <right paren>
 <interval qualifier>
 <interval value expression 1> ::= <interval value expression>

<interval term 1> ::= <interval term>
 <datetime term> ::= <datetime factor>
 <datetime factor> ::= <datetime primary>
 <datetime primary> ::= <value expression primary> | <datetime value function>
 <string value expression> ::= <character value expression>
 <null specification> ::= NULL
 <row value constructor list> ::= <row value constructor element> [{ <comma> <row value constructor element> } ...]
 <row subquery> ::= <subquery>
 <comp op> ::=
 <equals operator>
 | <not equals operator>
 | <less than operator>
 | <greater than operator>
 | <less than or equals operator>
 | <greater than or equals operator>
 <between predicate> ::=
 <row value constructor> [NOT] BETWEEN <row value constructor> AND <row value constructor>
 <in predicate> ::= <row value constructor> [NOT] IN <in predicate value>
 <in predicate value> ::= <table subquery> | <left paren> <in value list> <right paren>
 <in value list> ::= <value expression> { <comma> <value expression> } ...
 <like predicate> ::= <match value> [NOT] LIKE <pattern>
 <match value> ::= <character value expression>
 <pattern> ::= <character value expression>
 <escape character> ::= <character value expression>
 <null predicate> ::= IS [NOT] NULL
 <exists predicate> ::= EXISTS <table subquery>
 <row value constructor 1> ::= <row value constructor>
 <row value constructor 2> ::= <row value constructor>
 <truth value> ::= TRUE | FALSE
 <order by clause> ::= ORDER BY <sort specification list>
 <sort specification list> ::= <sort specification> [{ <comma> <sort specification> }...]

<sort specification> ::= <sort key> [<ordering specification>]

<sort key> ::= <column name> | <unsigned integer>

<ordering specification> ::= ASC | DESC